Počáteční úloha

Při simulace vývoje systému v čase používáme jednoduché zásady:

- Spojitý čas nahradíme posloupností časových okamžiků t_0, t_1, t_2, t_3, …, t_i, …. Interval mezi následujícími časovými okamžiky volíme obvykle konstantní ($\Delta t = t_{i+1} - t_i$), nazýváme ho *časový krok* a označujeme symbolem h.
- Stav systému je v každém čase popsán sadou (vektorem) stavových veličin, které plně určují následující vývoj systému.
- Výpočet nového stavu je závislý na bezprostředně předcházejícím stavu (může jich být i více, zde se však omezíme na jeden).

Abychom získali posloupnost následujících stavů představujících vývoj systému, musíme definovat nějaký výchozí stav, od kterého budeme další výpočet odvozovat (počáteční podmínku) – odtud název *počáteční úloha*.

Např. pohyb tělesa v homogenním gravitačním poli je popsán stavem (x, y, v_x, v_y), tj. složkami jeho souřadnic a rychlostí v rovině. Všechny složky stavového vektoru se v čase mění. V rámci uvedeného třetího bodu můžeme stav v čase t_i vyjádřit v závislosti na předcházejícím stavu obecně takto:

$$
\begin{aligned}
x_{i+1} &= f_1(x_i, y_i, v_{x,i}, v_{y,i}) \\
y_{i+1} &= f_2(x_i, y_i, v_{x,i}, v_{y,i}) \\
v_{x,i+1} &= f_3(x_i, y_i, v_{x,i}, v_{y,i}) \\
v_{y,i+1} &= f_4(x_i, y_i, v_{x,i}, v_{y,i})
\end{aligned}
$$

Funkce $f_1 \ldots f_4$ tvoří složky tzv. *vektorové funkce* a mohou být v našem případě triviální:

$x_{i+1} = x_i + v_{x,i}.h$	změnu souřadnic v jednom kroku aproximujeme rovnoměrně zrychleným pohybem; ve směru osy x je zrychlení nulové
$y_{i+1} = y_i + v_{y,i}.h - g.h^2/2$	ve směru osy y existuje zrychlení $-g$
$v_{x,i+1} = v_{x,i}$	složka rychlosti v ose x se nemění, v tomto směru nepůsobí žádná síla
$v_{y,i+1} = v_{y,i} - g.h$	rychlost ve směru osy y klesá působením gravitační síly

Připravíme si tedy následující list (zobrazené vzorce odrážejí přepis výše uvedených rovnic):

	A	B	C	D	E	F
1			Počáteční úloha			
2						
3	**krok** [s]	0,1				←Parametry úlohy
4	**g** [m/s²]	9,81				
5						
6	**t** [s]	**x** [m]	**y** [m]	**v_x** [m/s]	**v_y** [m/s]	← Stavové veličiny
7	0	0	0	10	10	← Počáteční podmínky
8	`=A7+$B$3`	`=B7+D7*$B$3`	`=C7+E7*$B$3-0,5*$B$4*$B$3*$B$3`	`=D7`	`=E7-$B$4*$B$3`	← Výpočet dalšího stavu
9						
10						

Adresy s parametry úlohy musí být pochopitelně absolutní, nechceme, aby se při kopírování vzorců měnily.

Nyní řádek vzorců zkopírujeme do následujících řádků:

	A	B	C	D	E	F
1	Počáteční úloha					
2						
3	**krok** [s]	0,1				←Parametry úlohy
4	**g** [m/s²]	9,81				
5						
6	**t** [s]	**x** [m]	**y** [m]	v_x [m/s]	v_y [m/s]	← Stavové veličiny
7	0	0	0	10	10	← Počáteční podmínky
8	0,1	1	0,95095	10	9,019	← Výpočet dalšího stavu
9						
10						
11						
12						
13						
14						
15						
16						
17						
18						
19						
20						

Z výsledných hodnot ve sloupci *x* a *y* sestrojíme bodový graf dráhy tělesa:

	A	B	C	D	E	F
1	Počáteční úloha					
2						
3	**krok** [s]	0,1				←Parametry úlohy
4	**g** [m/s²]	9,81				
5						
6	**t** [s]	**x** [m]	**y** [m]	v_x [m/s]	v_y [m/s]	← Stavové veličiny
7	0	0	0	10	10	← Počáteční podmínky
8	0,1	1	0,95095	10	9,019	← Výpočet dalšího stavu
9	0,2	2	1,8038	10	8,038	
10	0,3	3	2,55855	10	7,057	
11	0,4	4	3,2152	10	6,076	
12	0,5	5	3,77375	10	5,095	
13	0,6	6	4,2342	10	4,114	
14	0,7	7	4,59655	10	3,133	
15	0,8	8	4,8608	10	2,152	
16	0,9	9	5,02695	10	1,171	
17	1	10	5,095	10	0,19	
18	1,1	11	5,06495	10	-0,791	
19	1,2	12	4,9368	10	-1,772	
20	1,3	13	4,71055	10	-2,753	
21	1,4	14	4,3862	10	-3,734	
22	1,5	15	3,96375	10	-4,715	
23	1,6	16	3,4432	10	-5,696	
24	1,7	17	2,82455	10	-6,677	
25	1,8	18	2,1078	10	-7,658	
26	1,9	19	1,29295	10	-8,639	
27	2	20	0,38	10	-9,62	



Z vypočtených hodnot *y* můžeme pomocí funkce *MAX* nalézt maximální výšku vrhu:



B5 | =MAX(C9:C50)

	A	B	C	D	E	F	G	H
1		Počáteční úloha						
2								
3	**krok** [s]	0,1				←**Parametry úlohy**		
4	**g** [m/s^2]	9,81						
5	**výška** [m]	5,095	$v_y^2/(2.g)$ =	5,09683996				
6								
7								
8	**t** [s]	**x** [m]	**y** [m]	v_x [m/s]	v_y [m/s]	← **Stavové veličiny**		
9	0	0	0	10	10	← **Počáteční podmínky**		
10	0,1	1	0,95095	10	9,019	← **Výpočet dalšího stavu**		
11	0,2	2	1,8038	10	8,038			
12	0,3	3	2,55855	10	7,057			
13	0,4	4	3,2152	10	6,076			

Vidíme, že nalezené maximum dobře koresponduje s hodnotou vyplývající ze zákona zachování energie $y_{max} = v_y^2/(2g)$, kde v_y je vertikální složka počáteční rychlosti.

Pokročilejší řešení se zahrnutím odporu prostředí

Šikmý vrh je samozřejmě triviální úloha, která je analyticky řešitelná (parabola). Přidáme-li ke gravitační síle $\boldsymbol{G}$ působící na hmotný bod ještě odpor prostředí, situace se podstatně změní. Analytické řešení úlohy již v takovém případě neexistuje a musíme použít numerický výpočet, jehož základy jsme si vybudovali.

Odpor prostředí se projevuje silou $\boldsymbol{T}(\boldsymbol{v})$, která působí proti směru pohybu a je závislá na velikosti rychlosti. Pro výslednici sil působících na těleso platí:

$$\boldsymbol{G} + \boldsymbol{T}(\boldsymbol{v}) = m\boldsymbol{a}$$

Obvykle předpokládáme, že síla odporu prostředí má tvar

$$\boldsymbol{T}(\boldsymbol{v}) = \mathrm{C}|\boldsymbol{v}|.\boldsymbol{v},$$

kde C je koeficient odporu prostředí (vždy záporný), který závisí na geometrii objektu a hustotě prostředí. Absolutní hodnota (velikost vektoru) rychlosti je samozřejmě:

$$|\boldsymbol{v}| = \sqrt{\boldsymbol{v}_x^2 + \boldsymbol{v}_y^2}$$

Podle výše uvedeného vztahu je pak

$$|\boldsymbol{T}| = \sqrt{\boldsymbol{T}_x^2 + \boldsymbol{T}_y^2} = \boldsymbol{C}|\boldsymbol{v}|^2$$

Vektor síly $\boldsymbol{T}$ je tedy orientován proti směru vektoru rychlosti a jeho velikost je úměrná čtverci velikosti rychlosti. V aerodynamice se ukazuje, že tento vztah platí v širokém rozsahu hodnot rychlostí a hustot prostředí.

Z uvedených čtyř vztahů získáme vektor okamžitého zrychlení tělesa o složkách:

$$\boldsymbol{a}_x = \frac{\boldsymbol{C}\boldsymbol{v}_x}{\boldsymbol{m}}\sqrt{\boldsymbol{v}_x^2 + \boldsymbol{v}_y^2}$$

$$\boldsymbol{a}_y = \frac{\boldsymbol{C}\boldsymbol{v}_y}{\boldsymbol{m}}\sqrt{\boldsymbol{v}_x^2 + \boldsymbol{v}_y^2} - \boldsymbol{g}$$

Předpis pro stanovení nového stavu z předcházejícího stavu bude nyní za předpokladu rovnoměrně zrychleného pohybu mezi dvěma stavy vypadat následovně:

$$x_{i+1} = x_i + v_{x,i}h + a_{x,i}h^2/2$$
$$y_{i+1} = y_i + v_{y,i}h + a_{y,i}h^2/2$$
$$v_{x,i+1} = v_{x,i} + a_{x,i}h$$
$$v_{y,i+1} = v_{y,i} + a_{y,i}h$$

Na pravé straně rovnic jsou vždy pouze složky původního stavu. Proto také odvozené složky zrychlení musí používat rychlost v původním stavu. Zřetelně zde vidíme hlavní ideu počáteční úlohy: v rámci jednoho kroku považujeme vybrané veličiny za konstantní (zde například zrychlení). Chyba, které se tím dopustíme, bude malá a při menším časovém kroku se rovněž zmenší (úloha konverguje).

Poznámka: zrychlení není v tomto případě stavovou veličinou – je totiž zcela závislé na jiných stavových veličinách (rychlosti).

Řešení

Parametry úlohy je třeba rozšířit o koeficient odporu prostředí a o hmotnost vrženého objektu (ta v případě bez odporu prostředí nehraje roli – viz Galileův zákon).

Pro přehlednost přestěhujeme data s jednotlivými stavy do samostatného listu *data* a parametry uložíme do listu pojmenovaného např. *balistika*, který může vypadat např. takto:



	A	B	C	D	E
1	Počáteční úloha				
2	krok	0,01	s		
3	g	9,81	m/s^2		
4	v_0	28	m/s		
5	elev. úhel	30	stupňů		
6	odpor	-0,1	[F = $C.v^2$]		
7	m	5	kg		
8	Vypočtené údaje				
9	dolet		m		
10	výška		m		
11					
12					
13					
14					

Do listu jsme přidali ještě počáteční velikost a úhel rychlosti, což nám umožní komfortně měnit počáteční podmínky úlohy. Protože budeme parametry používat ve vzorcích, bývá dobrým zvykem je pojmenovat – vzorce jsou pak mnohem přehlednější. Pomocí správce názvů (záložka vzorce) tedy přiřadíme jednotlivým buňkám parametrů názvy např. takto:



List *data* si nyní připravíme v následující podobě:

	A	B	C	D	E
1	t	x	y	v_x	v_y
2	0	0	0	=v0*COS(uhel*PI()/180)	=v0*SIN(uhel*PI()/180)
3					

Počáteční podmínky zůstali obdobné, pouze složky počátečního vektoru rychlosti vypočítáme z velikosti a úhlu (z polárních souřadnic) pomocí známých vztahů (pozor, úhel ve stupních je třeba převést na radiány). Všimněte si také zlepšení čitelnosti vzorců využitím názvů buněk.

Sloupce dat rozšíříme o položky okamžitého zrychlení, které budou naplněny vzorci pro výpočet složek a_x a a_y z počáteční rychlosti a parametrů úlohy (viz výše):

	F	G
1	a_x	a_y
2	=odpor/m*ODMOCNINA(D2*D2+E2*E2)*D2	=odpor/m*ODMOCNINA(D2*D2+E2*E2)*E2-g
3		
4		

V dalším řádku definujeme vzorce pro výpočet nového stavu pomocí dat předcházejícího stavu s využitím definovaných názvů a vypočtených hodnot zrychlení podle předcházejícího obrázku:

	A	B	C	D	E
1	t	x	y	v_x	v_y
2	0	0	0	=v0*COS(uhel	=v0*SIN(uhel
3	=A2+h	=B2+D2*h+0,5*F2*h*h	=C2+E2*h+0,5*G2*h*h	=D2+F2*h	=E2+G2*h
4					
5					

Hodnoty zrychlení jsou závislé pouze na položkách rychlosti v tomtéž řádku, proto je lze pouze zkopírovat pomocí úchytu pro kopírování vzorců:

	F	G
1	a_x	a_y
2	=odpor/m*ODMOCNINA(D2*D2+E2*E2)*D2	=odpor/m*ODMOCNINA(D2*D2+E2*E2)*E2-g
3	=odpor/m*ODMOCNINA(D3*D3+E3*E3)*D3	=odpor/m*ODMOCNINA(D3*D3+E3*E3)*E3-g
4		

Máme tedy připraven celý řádek se vzorci pro výpočet nového stavu, opět ho zkopírujeme pomocí úchytu tentokrát do většího počtu řádků (např. 1000):

A3 fx =A2+h

	A	B	C	D	E	F	G	H
1	t	x	y	v_x	v_y	a_x	a_y	
2	0	0	0	24,24871	14	-13,579278	-17,65	
3	0,01	0,2418081	0,1391175	24,11292	13,8235	-13,404024	-17,49428	
4								
5								
6								
7								
8								
9								

Velký počet stavů potřebujeme proto, že jsme zvolili relativně malý časový krok (0,01s). Zároveň si vytvoříme rezervu pro nastavení libovolných parametrů úlohy – je třeba si uvědomit, že v naší konstrukci je počet vypočtených stavů pevný, i když se jejich obsah mění. Samozřejmě můžeme kdykoli zasáhnout do listu s daty a doplnit je kopírováním o nové stavy, z hlediska komfortu bychom však chtěli, abychom mohli manipulovat pouze s listem s parametry.

Do listu *balistika* nyní vložíme bodový graf, ve formuláři pro výběr dat přitom můžeme z listu *data* vybrat celé sloupce s hodnotami *x* a *y* (klikneme na názvy sloupců B a C). Graf si automaticky vybere jen ty buňky, které obsahují nějaké hodnoty:



Graf bude vypadat následovně:



Předpokládejme, že nás zajímají jen data před dopadem na zem, tj. stavy, kde $y \geq 0$. Upravíme proto svislou osu grafu ve formuláři pro vlastnosti hlavní svislé osy následovně:



V grafu se nyní potlačí záporné hodnoty ve svislé ose a pro parametry zobrazené v listu bychom měli dostat následující graf, který se při změně parametrů automaticky změní:



Pro automatický výpočet maximální výšky letu použijeme opět jednouchý vzorec pro nalezení maximální hodnoty v celém sloupci C s hodnotami *y* pro všechny stavy *=MAX(data!C:C)*. Tentokrát však již nebude platit teoretický vztah pro maximální výšku jako v předcházejícím případě se zanedbáním odporu prostředí. Počáteční kinetická energie bude částečně spotřebována na překonání odporu prostředí a těleso vystoupá do menší výšky.

Zjištění doletu (tj. velikosti hodnoty *x* v okamžiku dopadu na zem) použijeme následující postup:

Nejprve zjistíme číslo řádku v listu *data*, kde je souřadnice *y* nejmenší nezáporná. K tomu slouží vyhledávací funkce *POZVYHLEDAT (co; kde; jak)*. První parametr *co* obsahuje hledanou hodnotu (v našem případě velmi malé záporné číslo, např. -0,0000001). Druhý parametr *kde* obsahuje oblast buněk, v níž hodnotu hledáme (v našem případě je to celý sloupec C). Třetím parametrem konečně říkáme, zda v případě, že se nenajde přesně hledaná hodnota, použijeme nejbližší vyšší nebo nižší hodnotu (v našem případě je hodnota parametru -1, tj. použijeme nejbližší vyšší). Pokud tedy do libovolné buňky našeho sešitu vložíme vzorec *=POZVYHLEDAT(-0,0000001;data!C:C;-1)* bude vrácená hodnota např. 223, tj. číslo řádku s poslední nezápornou hodnotou ve sloupci C:



	A	B	C	D	E	F	G
1	t	x	y	v_x	v_y	a_x	a_y
221	2,19	14,691658	0,2016835	2,594041	-8,07608	-2,2003841	-2,959501
222	2,2	14,717488	0,1207747	2,572037	-8,10567	-2,1872496	-2,916969
223	2,21	14,743099	0,0395722	2,550164	-8,13484	-2,1740658	-2,874885
224	2,22	14,768492	-0,04192	2,528424	-8,16359	-2,160836	-2,833249
225	2,23	14,793668	-0,1236976	2,506815	-8,19192	-2,1475633	-2,792062
226	2,24	14,818629	-0,2057564	2,48534	-8,21984	-2,1342509	-2,751323

Pozorný čtenář si jistě všiml, že nemůžeme hledat hodnotu *0*, protože ta je obsažena hned v prvním řádku dat (počáteční podmínce).

Dolet tělesa je obsažen ve sloupci B na stejném řádku. Funkce, která vrací požadovanou buňku v požadovaném řádku se nazývá *INDEX(kde, číslo)*. Parametr *kde* definuje oblast (v našem případě sloupec B s hodnotami *x*), parametr *číslo* definuje číslo řádku (resp. sloupce podle toho, s jakou oblastí pracujeme). Vzorec *=INDEX(data!B:B;223)* tedy podle předcházejícího obrázku vrátí hodnotu 14,743099.

Výsledný vzorec, který musíme vložit do buňky s hledaným doletem, dostaneme tedy vložením funkce *POZVYHLEDAT* na místo parametru funkce *INDEX* podle obrázku:



Poznámka: Funkce *POZVYHLEDAT* se v původní verzi Excelu nazývá *MATCH*, což je evidentně výstižnější název – snaha o počeštění je zde značně kontraproduktivní.

Nyní si můžeme změnou parametrů snadno ověřit, že hmotnější těleso doletí vlivem větší setrvačnosti do větší vzdálenosti, také elevační úhel pro maximální dolet již není 45 stupňů a závisí na mnoha faktorech.

Pro vyhledání optimálního elevačního úhlu pro maximální dolet při dané počáteční rychlosti můžeme využít nástroj *Řešitel*, popsaný v předcházejícím příkladu:



Vzhledem k tomu, že od hodnoty počátečního úhlu k hodnotě doletu existují tisíce provázaných vzorců, nemusí řešitel nalézt hledaný optimální úhel. Ve formuláři *Řešitele* pak lze u metody (GRG nonlinear) pomocí tlačítka *Možnosti* zadat přesnější hodnoty konvergence a vyžadovat vyzkoušení většího počtu počátečních podmínek:



Nalezené řešení bychom vždy měli ověřit jinou úvahou,


A
B
C
Počáteční úloha
krok
0,01
s
g
9,81
m/s2
v0
28
m/s
elev. úhel
32,4349916
stupňů
odpor
-0,1
[F = C.v2]
m
1
kg
Výsledky Řešitele
Řešitel došel k aktuálnímu řešení, které splňuje všechny omezující podmínky.
Sestavy
Uchovat řešení Řešitele
Obnovit původní hodnoty
Výsledková
Citlivostní
Limitní
Zpět do dialogového okna Parametry Řešitele
Stručné sestavy
OK
Storno
Uložit scénář...
Řešitel došel k aktuálnímu řešení, které splňuje všechny omezující podmínky.